
Figure X1: Tel Azekah phase S2-6
[bookmark: _Toc306467406][bookmark: _Toc306467576][bookmark: _Toc307169319][image: ]
Figure X2: Tel Azekah phase S2-5b

Figure X3: Tel Azekah phase S2-5a
[image: ]
Figure X4: Tel Azekah phase S2-4

Figure X5: Tel Azekah phase T2-3

Figure X6: Curve plot of individual calibrated dates from Tel Azekah Area S2 Late Bronze phases.

[bookmark: _GoBack]Figure X7: Bayesian model following the initial stratigraphic association of OZS882 and OZS883 with Phase S2-6, and excluding the more recently dated sample (cluster OZU045/046/052). (Refer also to Webster 2015: 73-74, but note that phase numbering has since been updated).

Figure X8: Narrow span of S2-6 through S2-5a under the original stratigraphic association of OZS882 and OZS883 with phase S2-6 (model shown in Figure X7).

Figure X9: Modelled interval between S2-6 and S2-5b (Area S2 Model A, Figure 7). 

Table X1: Late Bronze phases and transitions at Lachish according to two models: 1) short-lived samples only (Model A, Figure 8); 2) all data (Model B, Figure 9).
 
	Boundary or Phase
	Cultural Period
	1σ range (BCE)
	2σ range (BCE)

	
	
	Model A
	Model B
	Model A
	Model B

	End VI
	
	1187-1092

	1155-1069
	1209-1015
	1186-1017

	VI
	LB III
	1192-1110
	1166-1090
	1211-1067
	1191-1048

	Transition VII to VI
	
	1203-1128
	1174-1119
	1224-1094
	1200-1088

	VII
	LB IIB
	1218-1153
	1196-1135
	1253-1117
	1228-1105

	Start VII
	
	1237-1162
	1216-1151
	1286-1124
	1253-1120

	End S-2
	
	
	1278-1188
	
	1330-1151

	S-2
	LB IIA
	
	1305-1212
	
	1354-1177

	Transition S-3 to S-2
	
	1296-1194
(End S-3)
	1336-1238
	1373-1156
(End S-3)
	1378-1200

	S-3
	LB IIA
	1371-1209
	1375-1256
	1427-1163
	1426-1216

	Start S-3
	
	1383-1225
	1401-1281
	1488-1138

	1468-1224





OXCAL MODELS

Tel Azekah – Area S2 Model A

Options()
 {
  Resolution=1;
 };
 Plot()
 {
  Outlier_Model("General",T(5),U(0,3),"t");
  Sequence("TEL AZEKAH LATE BRONZE PHASES, AREA S2")
  {
   Boundary("Start S2-6");
   Phase("Phase S2-6 (LB IIA)")
   {
    Label("(Three measurements from a cluster.)");
    R_Date("OZU045", 3090, 30)
    {
     Outlier("General",0.05);
    };
    R_Date("OZU046", 3060, 25)
    {
     Outlier("General",0.05);
    };
    R_Date("OZU052", 3055, 30)
    {
     Outlier("General",0.05);
    };
   };
   Boundary("End S2-6");
   Interval("End S2-6 to Start S2-5b");
   Boundary("Start S2-5b");
   Phase("Phase S2-5b (LB IIB)")
   {
    R_Date("OZS874", 2985, 25)
    {
     Outlier("General",0.05);
    };
    R_Date("OZS881", 2915, 25)
    {
     Outlier("General",0.05);
    };
   };
   Boundary("Transition S2-5b to S2-5a");
   Phase("Phase S2-5a (LB IIB)")
   {
    R_Date("OZS873", 2965, 30)
    {
     Outlier("General",0.05);
    };
    R_Date("OZS880", 2960, 30)
    {
     Outlier("General",0.05);
    };
    R_Date("OZS875", 2985, 30)
    {
     Outlier("General",0.05);
    };
   };
   Boundary("End S2-5a");
  };
};


Tel Azekah – Area S2 Model B 
(elimination of OZS881 using agreement indices rather than outlier analysis)

Options()
 {
  Resolution=1;
 };
 Plot()
 {
  Sequence("TEL AZEKAH LATE BRONZE PHASES, AREA S2")
  {
   Boundary("Start S2-6");
   Phase("Phase S2-6 (LB IIA)")
   {
    Label("(Three measurements from a cluster.)");
    R_Date("OZU045", 3090, 30);
    R_Date("OZU046", 3060, 25);
    R_Date("OZU052", 3055, 30);
   };
   Boundary("End S2-6");
   Interval("End S2-6 to Start S2-5b");
   Boundary("Start S2-5b");
   Phase("Phase S2-5b (LB IIB)")
   {
    R_Date("OZS874", 2985, 25);
   };
   Boundary("Transition S2-5b to S2-5a");
   Phase("Phase S2-5a (LB IIB)")
   {
    R_Date("OZS873", 2965, 30);
    R_Date("OZS880", 2960, 30);
    R_Date("OZS875", 2985, 30);
   };
   Boundary("End S2-5a");
  };
};


Tel Azekah – Area T2

Options()
 {
  Resolution=1;
 };
 Plot()
 {
  Outlier_Model("General",T(5),U(0,4),"t");
  Sequence("TEL AZEKAH LATE BRONZE PHASES, AREA T2")
  {
   Boundary("Start T2-3b");
   Phase("T2-3b (LB III)")
   {
    R_Combine("Av. OZV243, 52509")
    {
     Outlier("General",0.05);
     R_Date("52509",2953, 24);
     R_Date("OZV243", 2925, 30);
    };
   };
   Boundary("Transition T2-3b to T2-3a");
   Phase("T2-3a (LB III)")
   {
    R_Combine("Av. OZV265, 52510")
    {
     Outlier("General",0.05);
     R_Date("52510",2955, 23);
     R_Date("OZV265", 2890, 25);
    };
   };
   Boundary("End T2-3a");
  };
};


Tel Lachish Model A (short-lived samples, Area S)

Options()
 {
  Resolution=1;
 };
 Plot()
 {
  Outlier_Model("General",T(5),U(0,3),"t");
  Sequence("LACHISH LATE BRONZE LEVELS (short-lived samples, Area S)")
  {
   Boundary("Start S3");
   Phase("Level S-3 (LBIIA)")
   {
    R_Date("RT-3152", 2945, 65)
    {
     Outlier("General", 0.05);
    };
    R_Date("RT-3153", 3125, 55)
    {
     Outlier("General", 0.05);
    };
   };
   Boundary("End S3");
   Label("Unspecified gap allowed for Levels S2 through VIIb")
   Interval("End S3 to Start VIIa");
   Boundary("Start VIIa");
   Phase("Level VIIa (LBIIB)")
   {
    R_Date("Pta-3320", 2940, 70)
    {
     Outlier("General", 0.05);
    };
    R_Date("RT-2906", 2955, 35)
    {
     Outlier("General", 0.05);
    };
   };
   Boundary("Transition VII to VI");
   Phase("Level VI (LB III)")
   {
    R_Date("RT-2912", 2915, 25)
    {
     Outlier("General", 0.05);
    };
    R_Date("RT-2755", 2955, 25)
    {
     Outlier("General", 0.05);
    };
   };
   Boundary("End VI");
  };
};


Tel Lachish Model B (wood charcoal and short-lived samples)
Note: For the Charcoal Plus model data, refer to Dee and Bronk Ramsey (2014, SOM).

Options()
 {
  Resolution=1;
 };
 Plot()
 {
  Outlier_Model("General",T(5),U(0,4),"t");
  Outlier_Model("Charcoal_p",Prior("Charcoal_p.prior"),U(0,3),"t");
  Sequence("LACHISH LATE BRONZE LEVELS (wood charcoal and short-lived samples)")
  {
   Boundary("Start S-3");
   Phase("Area S Level S-3 (LB IIA)")
   {
    R_Date("RT-2756", 3080, 25)
    {
     Outlier("Charcoal_p", 1);
    };
    R_Date("RT-2916", 3210, 25)
    {
     Outlier("Charcoal_p", 1);
    };
    R_Date("RT-3152 (s)", 2945, 65)
    {
     Outlier("General", 0.05);
    };
    R_Date("RT-3153 (s)", 3125, 55)
    {
     Outlier("General", 0.05);
    };
   };
   Boundary("Transition S-3 to S-2");
   Phase("Area S Level S-2 (LB IIA)")
   {
    R_Date("RT-2754 (s)", 3945, 25)
    {
     Outlier("General", 0.05);
    };
    R_Date("RT-2918", 3090, 40)
    {
     Outlier("Charcoal_p", 1);
    };
   };
   Boundary("End S-2");
   Label("Unspecified gap allowed for Area S Level S-1");
   Interval();
   Boundary("Start VII");
   Phase("General Level VII (LB IIB)")
   {
    Label("Area S Level VIIa");
    R_Date("RT-3154", 2955, 25)
    {
     Outlier("Charcoal_p", 1);
    };
    R_Date("Pta-3320", 2940, 70)
    {
     Outlier("Charcoal_p", 1);
    };
    R_Date("RT-2906 (s)", 2955, 35)
    {
     Outlier("General", 0.05);
    };
    R_Date("RT-2757", 2995, 25)
    {
     Outlier("Charcoal_p", 1);
    };
    R_Date("RT-3155", 3055, 20)
    {
     Outlier("Charcoal_p", 1);
    };
    R_Date("RT-2910", 2885, 35)
    {
     Outlier("Charcoal_p", 1);
    };
    R_Date("RT-3156", 3025, 25)
    {
     Outlier("Charcoal_p", 1);
    };
    Label("Area P Level P-1");
    R_Date("RT-2913", 3855, 40)
    {
     Outlier("General",0.05);
    };
   };
   Boundary("Transition VII to VI");
   Phase("General Level VI (LB III)")
   {
    Label("Area S (Pillared Building)");
    R_Date("RT-2912 (s)", 2915, 25)
    {
     Outlier("General", 0.05);
    };
    R_Date("RT-2755 (s)", 2955, 25)
    {
     Outlier("General", 0.05);
    };
    R_Date("RT-2914", 2985, 30)
    {
     Outlier("Charcoal_p", 1);
    };
    Label("Area GE (Inner City Gate)");
    R_Date("RT-2920", 2850, 40)
    {
     Outlier("Charcoal_p", 1);
    };
    R_Date("RT-3157", 2955, 25)
    {
     Outlier("Charcoal_p", 1);
    };
    Label("Area P (Temple)");
    R_Date("Hel-810", 3090, 120)
    {
     Outlier("Charcoal_p", 1);
    };
    R_Date("Hel-1421", 3000, 100)
    {
     Outlier("Charcoal_p", 1);
    };
    R_Date("Hel-1417 (s)", 2810, 100)
    {
     Outlier("General", 0.05);
    };
    R_Date("Pta-3370", 3090, 60)
    {
     Outlier("Charcoal_p", 1);
    };
    R_Date("Hel-1028", 3510, 120)
    {
     Outlier("Charcoal_p", 1);
    };
   };
   Boundary("End VI");
  };
};

